
//****************************************************
// Copyright (C) 1997 Maximizer Technologies Inc.
//****************************************************
//InsertMergeField
Application ( A1; "WordPerfect"; Default; "US" )

global UDFFileName
UDFFileName = GetUDFFile()

if ( not MaximizerIsLoaded() ) go Exit_InsertField endif
if ( not DatabaseIsOpen() ) go Exit_InsertField endif
if ( not LoopUntilMaximizerIsReady() ) go Exit_InsertField endif
if ( ListIsEmpty() ) go Exit_InsertField endif
if ( PersonalHotList() ) go Exit_InsertField endif

if ( DisplayDialog( &field; &link ) )
DoInsert( field; link)

endif

Label(Exit_InsertField)

quit

//********************************************************
// inserts the field as either a DDE link or as plain text
//********************************************************
procedure DoInsert( field; link )

if (not LoopUntilMaximizerIsReady() ) go Exit_DoInsert endif

if (link)
OnError(WrongPosition)
link_Name = "Max" + numStr(?Random)
DDECreateLink( LinkName: link_Name; Source: "MAXWIN|

CurrentRecord|" + field; UpdateMode: Automatic!; Format: TEXT!)

DDEUpdateLink( link_Name )
else

ddeCurrent = InitiateMaxWinDDE("CURRENTRECORD")
if (ddeCurrent <> 0)

OnError( InsertFieldTerminateDDE )

contents = ddeRequest( ddeCurrent; field)
type( contents)

Label( InsertFieldTerminateDDE )
ddeTerminate( ddeCurrent )

endif
endif
Go (Exit_DoInsert)

Label (WrongPosition)
MessageBox(i; "Maximizer Template"; "You may be attempting to 

insert a link inside another link. Please move the cursor to a different place
and try this operation again."; Ok! | IconExclamation!)

Go (Exit_DoInsert)



Label(Exit_DoInsert)
endproc

//********************************************************
// displays the Insert Maximizer Field dialog
//********************************************************
function DisplayDialog( &field; &link )

DialogDefine (Dialog:"InsertMaximizerField"; Left:50; Top:50; Width:195;
Height: 128; Style:Percent!; Caption:"Insert Merge Field")

DialogAddText (Dialog:"InsertMaximizerField"; Control:"Static1"; Left:5;
Top:5; Width:82; Height:10; Style:Left!; Text:"&Select a Merge Field:")

DialogAddListBox (Dialog:"InsertMaximizerField"; Control:"lstFields"; 
Left:5; Top:14; Width: 135; Height:109; Style:WPChars! | Sorted!; MacroVar: 
lstFields)

DialogAddPushButton (Dialog:"InsertMaximizerField"; Control:"btnOK"; 
Left:149; Top:14; Width: 38; Height:13; Style:DefaultBttn!; ButtonText:"OK")

DialogAddPushButton (Dialog:"InsertMaximizerField"; Control: 
"btnCancel"; Left:149; Top:31; Width: 38; Height:13; Style: CancelBttn!; 
ButtonText: "Cancel")

DialogAddPushButton( Dialog: "InsertMaximizerField"; Control: "btnHelp";
Left:149; Top:49; Width: 38; Height:13; Style: NonDefaultBttn!; 
ButtonText:"Help")

DialogAddCheckBox (Dialog:"InsertMaximizerField"; Control:"chkLink"; 
Left:149; Top:65; Width:50; Height:10; Text:"&Link"; MacroVar: chkLink)

AddAddressFields
if (ListUDFsToFile()) ReadUDFs endif

WM_SYSCOMMAND = 274
WM_COMMAND = 273
dlgResult = 999
chkLink = 1
DialogShow ( "InsertMaximizerField"; "WordPerfect"; HandleDialog; 

"lstFields")

while( dlgResult = 999)
endWhile
go( Exit_DisplayDialog )

Label(HandleDialog)
if (HandleDialog[5] = WM_COMMAND)

if (HandleDialog[3] = "btnOK") 
RegionGetSelectedText( field; 

"InsertMaximizerField.lstFields")
if (field <> "") dlgResult = -1 endif

endif
if (HandleDialog[3] = "btnCancel") dlgResult = 0 endif

if (HandleDialog[3] = "btnHelp") DisplayHelp(58000) endif
endif
if (HandleDialog[5] = WM_SYSCOMMAND) dlgResult = 0 endif
return

Label( Exit_DisplayDialog )



RegionGetSelectedText( field; "InsertMaximizerField.lstFields")
field = ReplaceWithIn( " "; "_"; field)
RegionGetCheck( link; "InsertMaximizerField.chkLink")

// DialogDestroy( Dialog: "InsertMaximizerField")  
DialogDismiss( "InsertMaximizerField"; "btnOK")
return ( dlgResult = -1 )

endfunc

//****************************************************
// adds the standard address fields to the listbox
//****************************************************
procedure AddAddressFields()
FOREACH (fieldName; {"Address Line 1"; "Address Line 2"; "City"; "Company"; 
"Country"; "Department"; "Division"; "Fax Number"; "First Name"; "Initial"; 
"Identification"; "Last Name"; "Mr/Ms"; "Name and Address"; "Phone 1"; "Phone 
1 Extension"; "Phone 2"; "Phone 2 Extension"; "Phone 3"; "Phone 3 Extension"; 
"Phone 4"; "Phone 4 Extension"; "Position"; "Salutation"; "State / Province"; 
"Today's Date"; "Today's Time"; "Zip / Postal Code"} )

DialogAddListItem("InsertMaximizerField";"lstFields"; fieldName)

ENDFOR
endproc

//****************************************************
// returns a file name to store the list of UDFs
//****************************************************
function GetUDFFile()

dllcall prototype GetTempPath( "kernel32.dll"; ; DWORD; {DWORD(bufSize);
address(ansistring(buffer))})

dllcall prototype GetTempFileNameA( "kernel32.dll"; integer; 
{ansistring(tmpPath); ansistring(prefix); unique; 
address(ansistring(buffer)) } )

// get the temporary directory
tmppath = "123456789012345678901234567890123456789012345678901234567890"
r = GetTempPath( 60; tmppath)

//get the temporary filename
zprefix = "UDF"
fileName= "123456789012345678901234567890123456789012345678901234567890"
r = GetTempFileNameA( tmpPath; zprefix; 0; filename)

return ( fileName )
endfunc

//****************************************************
// list all Maximizer UDFs to a file
//****************************************************
function ListUDFsToFile( )

result = false
onError( Exit_ListUDFsToFile)
ddeSystem := InitiateMaxWinDDE( "SYSTEM" )
if (ddeSystem <> 0)



OnError( ListUDFsTerminateDDE )
ddeExecute( ddeSystem; "[ListOfCategories(Headings," + UDFFileName

+ ")]" )

Label( ListUDFsTerminateDDE )
ddeTerminate( ddeSystem )
result = true

endif

Label(Exit_ListUDFsToFile)
return( result )

endfunc

//****************************************************
// read UDFs from file & add to the listbox
//****************************************************
procedure ReadUDFs()

hFile = OpenFile( UDFFileName )
if (hFile <= 0) go Exit_ReadUDFs endif

err = 0
while ( ( err >= 0 ) and (not FileIsEOF(hFile) ) )

err = FileRead( hfile; line)
// remove UDF type
cp = strLen( line )
while ((cp > 0) and (substr( line; cp; 1 ) <> ","))

cp = cp - 1
endwhile

line = substr( line; 1; cp - 1)

DialogAddListItem("InsertMaximizerField";"lstFields"; line)
endwhile
closeFile( hFile )

Label(Exit_ReadUDFs)
DeleteFile( UDFFilename; NoPrompts! ) 

endproc

////////////////////////////////////////////////////////////////////////
//
// Shared Functions

//********************************************************************
//returns true if Maximizer is loaded, false otherwise
//********************************************************************
function MaximizerIsLoaded()

isLoaded = false
onError( Exit_MaximizerIsLoaded )

isLoaded = ( AppLocate( "Maximizer*" ) <> 0 )

if ( not isLoaded )
messagebox( vReturn; "Maximizer Template"; "Please start 

Maximizer."; Ok! | IconExclamation!)
endif



Label(Exit_MaximizerIsLoaded)
return( isLoaded )

endfunc

//****************************************************
// Returns true if Maximizer is busy, false otherwise
//****************************************************
function MaximizerIsBusy()

returnValue = true
OnError( Exit_MaximizerIsBusy )

ddeCurrent = InitiateMaxWinDDE( "SYSTEM" )
if (ddeCurrent <> 0)

onError( MaximizerIsBusyDDETerminate )
returnValue :=  (DDERequest( ddeCurrent; "BUSYINDICATOR" ) = 

"BUSY")

Label( MaximizerIsBusyDDETerminate )
ddeTerminate( ddeCurrent )

endif

Label( Exit_MaximizerIsBusy )
return( returnValue )

endfunc

//*****************************************************
//loops until Maximizer is ready, or until User presses cancel
//returns false if user cancel, else returns true
//*****************************************************
function LoopUntilMaximizerIsReady()

vResult = 1
onError( Exit_LoopUntilMaximizerIsReady )

while ( ( vResult = 1 ) and ( MaximizerIsBusy() ) )
messagebox( vResult; "Maximizer Template"; "Maximizer is currently

busy with another task. Please complete all tasks within Maximizer and click 
OK to continue."; OKCANCEL! | IconExclamation!)

endwhile

Label( Exit_LoopUntilMaximizerIsReady )
return( vResult = 1 )

endfunc

//*****************************************************
//replaces all x characters with y in string z
//*****************************************************
function ReplaceWithIn( x; y; z)

out = ""
theLength = strlen(z)
for( lp; 1; lp <= theLength; lp+1 )

ch = substr( z; lp; 1 )
if (ch = x) ch = y endif
out = out + ch



endfor
return( out )

endfunc

//****************************************************
//starts a DDE conversation with MaxWin
//****************************************************
function InitiateMaxWinDDE(topic)

dde = 0
onError( Error_InitiateMaxWinDDE )

Label( Retry_InitiateMaxWinDDE )
dde = DDEInitiate( "MAXWIN"; topic )
if ( dde <> 0) go( Exit_InitiateMaxWinDDE ) endif

Label( Error_InitiateMaxWinDDE )
messagebox( vReturn; "Maximizer Template"; "Unable to initiate a 

conversation with Maximizer. Please verify that Maximizer is running, that a 
Maximizer database is open, and that no dialogs are currently active."; 
OKCANCEL! | IconExclamation!)

if ( vReturn = 1 )
go( Retry_InitiateMaxWinDDe )

endif

Label( Exit_InitiateMaxWinDDE )
return( dde )

endfunc

//****************************************************
//returns true if the current Maximizer list is empty
//****************************************************
function ListIsEmpty()

isEmpty = true
onError( Exit_ListIsEmpty )

ddeCurrent = InitiateMaxWinDDE( "CURRENTRECORD" )
if ( ddeCurrent = 0 ) go Exit_ListIsEmpty endif

onError( ListIsEmptyDDETerminate )
dummy = ddeRequest( ddeCurrent; "GetListCount" ) = "0"
isEmpty = false

Label( ListIsEmptyDDETerminate )
ddeTerminate( ddeCurrent )
if ( isEmpty ) messageBox( vResult; "Maximizer Template"; "The Maximizer

list is empty."; OK! | IconExclamation!) endif

Label( Exit_ListIsEmpty )
return ( isEmpty )

endfunc

//****************************************************
//displays help
//****************************************************



function DisplayHelp(contextID)
dllcall prototype WinHelpA("User32.dll"; ; integer; {hwnd; 

ansistring(helpfile); command; data})

r = winhelpA( 0;"maxwin.hlp"; 1; contextID)
endfunc

//////////////////////////////////////////////////////////////////////
//
//MMW: newly added functions
//*****************************************************
//returns true if the current entry does not have an ID
//*****************************************************
function PersonalHotList()

retValue = false

ddeCurrent = InitiateMaxWinDDE("CurrentRecord")
if (ddeCurrent <> 0)

if (ddeRequest( ddeCurrent; "Identification") = "")
retValue = true
messageBox( vDlg; "Maximizer Template"; "This action cannot 

be performed when a personal HotList task is selected."; OK! | 
IconExclamation!)

endif
ddeTerminate( ddeCurrent )

endif
return( retValue )

endfunc

//*****************************************************
//returns true if the current entry does not have an ID
//*****************************************************
function DatabaseIsOpen()

retValue = false

ddeSystem = InitiateMaxWinDDE( "SYSTEM" )
if (ddeSystem <> "")

dbPath = ddeRequest( ddeSystem; "CurrentDatabasePath")
if (dbPath <> "n/a")

retValue = true
else

messagebox( vDlg; "Maximizer Template"; "Please open a 
Maximizer Address Book Folder."; OK! | IconExclamation!)

endif
endif
return( retValue )

endfunc


